

# Pràctiques Integrades 1er de Matemàtiques

Pràctica 3

Curs 2002–03

## 3 Gràfics

En aquesta pràctica aprendreu a dibuixar el gràfic d'una funció definida per una expressió. A més, altres temes que hi podeu trobar inclouen: combinar gràfics de diferents expressions en un únic dibuix, representar punts, i combinar diferents estructures gràfiques en un únic dibuix.

```
> restart;
```

### 3.1 Representar una expressió: la comanda `plot( )`

La comanda bàsica per representar gràficament una expressió o funció d'una variable és `plot`. Aprenem el seu funcionament al realitzar els exemples següents.

#### Exemple 3.1


Utilitzem la comanda `plot( )` per a dibuixar el gràfic de  $3x^2 - 8$  per a  $x$  entre  $-5$  i  $5$ .

```
> plot(3*x^2-8,x=-5..5);
```

Observeu que Maple posa l'escala de l'eix de les  $y$  de manera automàtica, triant l'escala de les  $y$  que mostra tot el gràfic que correspon al domini que s'ha especificat.

Podem eliminar l'elecció automàtica de l'escala de les  $y$  especificant a part del rang de les  $x$  un rang per a les  $y$ . En la línia següent hem limitat el rang de les  $y$  a l'interval  $[-20, 40]$ .

```
> plot(3*x^2-8,x=-5..5,y=-20..40);
```

Si feu clic amb el botó esquerre del ratolí, el gràfic queda seleccionat i la barra d'opcions inferior es modifica. Ara quan feu clic en el gràfic, les coordenades del punt del lloc on ho feu es veuran en la finestra de l'esquerra. El botó  fa que les escales de les  $x$  i de les  $y$  siguin iguals.

Podem experimentar sobre el gràfic anterior les diferents opcions possibles.

#### Exemple 3.2

L'escala automàtica és una característica útil però a vegades necessitarem especificar manualment el rang de les  $y$ . Per exemple l'escalat automàtic no és apropiat per a gràfics amb asímptotes verticals.

Compareu els dos gràfics següents. Noteu que hem fixat els límits per a les  $y$  a l'interval  $[-20, 20]$  en la segona comanda `plot( )`.

```
> plot(x/(x-2),x=-5..5);
> plot(x/(x-2),x=-5..5,y=-20..20);
```

### Exemple 3.3

Dibuixarem el gràfic de  $y = x^3 + 1 - e^x$  en el domini  $[-8, 8]$ . Triarem un rang per a les  $y$  que permeti veure els quatre talls amb l'eix de les  $x$ .

Primer doneu un cop d'ull al gràfic amb l'escala automàtica per a les  $y$ .

```
> plot(x^3+1-exp(x),x=-8..8);
```

Com que els valors de les  $y$  a prop de 8 són negatius i molt grans en valor absolut l'escala vertical ha hagut de ser massa gran per a veure clarament els talls amb l'eix de les  $x$ .

S'obté una visió molt millor restringint els límits en el rang de les  $y$ .

```
> plot(x^3+1-exp(x),x=-8..8,y=-5..15);
```

### Exercici 3.1

Dibuixeu  $y = \sin(x)$  per a dos períodes complets.

### Exercici 3.2

Dibuixeu  $y = 3x^4 - 6x^2$  per al domini  $[-10, 10]$  amb escala automàtica per a les  $y$ . Després d'observar el gràfic, editeu el domini i el recorregut per tal de veure els talls amb l'eix de les  $x$  clarament. Feu una estimació dels talls amb l'eix  $x$  amb el cursor.

## 3.2 Representar diferents expressions

Per a mostrar més d'un gràfic en el mateix dibuix feu-ne la llista dins de claudàtors `[ ]` separant-los per comes. Per exemple,

```
> plot([cos(x),x^2],x=-1..4,y=-4..4);
```

Noteu que cada un dels gràfics es mostra utilitzant un color diferent. Podeu especificar els colors per a cada funció afegint una opció de color al final de la comanda. Els colors s'assignaran en el mateix ordre que el de les funcions. Fixeu-vos que la llista dels colors també es fa amb uns claudàtors [ ]. Aquí hi ha un exemple.

```
> plot([cos(x),x^2],x=-1..5,y=-4..4,color=[blue,black]);
```

Aquests són els colors disponibles en Maple (s'han d'escriure en anglès, no val posar **groc**):

aquamarine, black, blue, navy, coral, cyan, brown, gold, green, gray, grey, khaki, magenta, maroon, orange, pink, plum, red, sienna, tan, turquoise, violet, wheat, white i yellow.

### Exercici 3.3

Feu junts els gràfics de les funcions  $y = x^2 - 5x + 6$  i  $y = \frac{1}{(x-2)^2}$ . Experimenteu amb diferents rangs per a les  $y$  de forma que es puguin veure dibuixos complets dels dos gràfics.

### 3.3 Representar punts

La comanda `plot( )` pot dibuixar també un o més punts.

#### Exemple 3.4

Dibuixem el punt (2,3). Noteu en la línia següent que utilitzem dos jocs de claudàtors.

```
> plot([ [2,3] ],style=point);
```

Podem controlar el tamany dels rangs per a les  $x$  i per a les  $y$  afegint aquesta informació a la comanda com en la línia següent.

```
> plot([ [2,3] ],x=-7..7,y=-7..7,style=point);
```

#### Exemple 3.5

Per a dibuixar més d'un punt fem una llista dins la comanda `plot( )` (observeu les comes). Recordeu que s'ha de posar un parell de claudàtors per cada punt i un parell extra envoltant la llista.

```
> plot([ [2,3], [-2,5], [1,-4] ],x=-7..7,y=-7..7,style=point);
```

**Exemple 3.6**

Canviant l'estil a "line" es connecten els punts conservant l'ordre de la llista.

```
> plot([ [2,3], [-2,5], [1,-4] ], x=-7..7, y=-7..7, style=line);
```

Què passa si canviem l'ordre dels punts de la llista?

**Exemple 3.7**

Es poden utilitzar extensions opcionals per a especificar el color dels punts i el símbol que es fa servir (per exemple "diamond", "circle" i "cross", que és el que hi ha per defecte) per a representar-los. Utilitzant l'ajuda de Maple, esbrineu les diferents possibilitats que admet l'opció `symbol`.

```
> plot([[3,2], [-2,3], [2,-1]], style=point, color=blue, symbol=circle);
```

**Exercici 3.4**

Dibuixeu els punts següents utilitzant el color vermell i el símbol "diamond":  $[1, 4]$ ,  $[-2, -3]$ ,  $[4, -5]$  i  $[-6, 5]$ . Després connecteu els punts amb línies rectes amb una comanda `plot( )` a part.

**3.4 Combinar gràfics d'expressions i punts: la comanda `display( )`**

Hi ha comandes de Maple que no estan disponibles fins que no es *carrega* un *paquet* que les activa. Un d'aquests paquets és el **plots** que estén les possibilitats gràfiques. Per accedir a les comandes d'un paquet (carregar el paquet) s'utilitza la comanda `with( )`, la línia següent carrega el paquet **plots**

```
> with(plots);
```

Després d'executar aquesta comanda veureu que Maple ha mostrat, com a resultat, el nom de les comandes noves que proporciona el paquet. Si no volem que el resultat d'una comanda es mostri per pantalla però sí que s'executi, podem acabar la línia amb dos punts enlloc de punt i coma. Els dos punts al final de la línia fa que la línia es pugui executar sense mostrar cap resultat i serveix per a qualsevol comanda.



```
> with(plots):
```

De les comandes noves que apareixen quan carreguem **plots** mencionarem únicament la comanda `display( )` que permet combinar gràfics d'expressions i de punts i línies en el mateix dibuix. Per utilitzar `display( )` el primer pas consisteix en nomenar individualment cada un dels components del dibuix.

**Important:** Ens hem d'assegurar que utilitzem uns **dos punts** al final de cada línia per a suprimir la presentació dels resultats ja que si no es fa el que s'obté com a resultat és força desagradable (mireu les tres primeres línies que hi ha a sota).

Després la comanda `display( )` s'utilitza per a fer el dibuix que volem (acaba amb un punt i coma).

```
> pict1:=plot([-3*x+5,9-x^2],x=-3..5,color=[green,red]):  
> pict2:=plot([[ -1,8],[4,-7]],style=point,color=blue,symbol=circle):  
> display([pict1,pict2]);
```

De forma alternativa podem fer la llista d'aquestes tres comandes `plot( )` relacionades en un sol grup d'execució. Com que sovint sortiran línies bastant llargues és útil canviar de línia sense executar la comanda prement les teclcs  +  a l'hora en el punt on volem que acabi cada una de les línies.

```
> pict1:=plot([-3*x+5,9-x^2],x=-3..5,color=[green,red]):  
  pict2:=plot([[ -1,8],[4,-7]],style=point,color=blue,symbol=circle):  
  display([pict1,pict2]);
```

### Exercici 3.5

Feu un gràfic que contingui a l'hora el gràfic de la funció  $y = x^2 + x - 6$  i les seves interseccions amb l'eix de les  $x$  i el de les  $y$  (per a  $x = -3$  i  $x = 2$ ) marcades amb un cercle. (Podeu consultar en l'ajuda de Maple com la comanda `solve( )`, que introduïrem en una pràctica propera, us pot donar aquests valors de la  $x$ ).